
Change 1 – Correction to function ‘f_GERAN_CSFB_Redirection_CCO_MT_Steps5_34()’
	Function name
	f_GERAN_CSFB_Redirection_CCO_MT_Steps5_34

	Reason for change
	In function f_GERAN_CSFB_Redirection_CCO_MT_Steps5_34, the Logical Channel type value returned by function f_GERAN_EnterU10_MT_Step10_19 should be used while calling functions f_GERAN_CC_Disconnect and f_G_ChannelRelease.

	Summary of change
	Declared a variable v_TCHType to capture the return value and passed it as a parameter in functions f_GERAN_CC_Disconnect and f_G_ChannelRelease.

	TTCN module
	MultiLayer_Procedures_GERAN.ttcn

	MCC160 Comment
	


Before:
	function f_GERAN_CSFB_Redirection_CCO_MT_Steps5_34 (GERAN_CellId_Type p_CellId,
                                                      G_LogicChType     p_ChannelType,
                                                      SubChannelNumber  p_SubChannel) runs on GERAN_PTC
  {
    //Authentication
    f_GERAN_MM_Authentication (p_CellId, p_ChannelType, p_SubChannel);
    //Activate Ciphering
    f_GERAN_Ciphering_Mode_Setting (p_CellId, tsc_PhyCh0, p_ChannelType, p_SubChannel);
    //Steps 10 to 19 of the generic test procedure in TS 51.010-1 subclause 10.1.3 are performed
    f_GERAN_EnterU10_MT_Step10_19(p_CellId, p_ChannelType, p_SubChannel);
    //Steps 20-22: Disconnect the MT call with MT procedure
    f_GERAN_CC_Disconnect(p_CellId, p_ChannelType, p_SubChannel);
    //Release channel
    f_G_ChannelRelease (p_CellId, tsc_PhyCh0, p_ChannelType, p_SubChannel);
    //Perform RAU
    f_GERAN_508CheckCampOnNewGeranCell(p_CellId);
  }


After:
	function f_GERAN_CSFB_Redirection_CCO_MT_Steps5_34 (GERAN_CellId_Type p_CellId,
                                                      G_LogicChType     p_ChannelType,
                                                      SubChannelNumber  p_SubChannel) runs on GERAN_PTC
  {
    //Declare a variable
    var G_LogicChType v_TCHType;
    //Authentication
    f_GERAN_MM_Authentication (p_CellId, p_ChannelType, p_SubChannel);
    //Activate Ciphering
    f_GERAN_Ciphering_Mode_Setting (p_CellId, tsc_PhyCh0, p_ChannelType, p_SubChannel);
    //Steps 10 to 19 of the generic test procedure in TS 51.010-1 subclause 10.1.3 are performed
    v_TCHType := f_GERAN_EnterU10_MT_Step10_19(p_CellId, p_ChannelType, p_SubChannel);
    //Steps 20-22: Disconnect the MT call with MT procedure
    f_GERAN_CC_Disconnect(p_CellId, v_TCHType, p_SubChannel);
    //Release channel
    f_G_ChannelRelease (p_CellId, tsc_TchCh, v_TCHType, p_SubChannel);
    //Perform RAU
    f_GERAN_508CheckCampOnNewGeranCell(p_CellId);
  }


Change 2 – Correction to function ‘f_GERAN_EnterU10_MT_Step10_19()’ 
	Function name
	f_GERAN_EnterU10_MT_Step10_19

	Reason for change
	According to TS 24.008 secn 9.3.2, Stream Identifier is a optional IE. Hence, in function f_GERAN_EnterU10_MT_Step10_19, template cr_G_CallConfirmed should be modified to reflect this.

	Summary of change
	Template cr_G_CallConfirmed is called with one parameter.

	TTCN module
	GERAN_CommonFunctions.ttcn

	MCC160 Comment
	


Before:
	function f_GERAN_EnterU10_MT_Step10_19 (GERAN_CellId_Type p_CellId,
                                          G_LogicChType p_ChannelType,
                                          SubChannelNumber p_SubChannel) runs on GERAN_PTC return G_LogicChType
  {
    var G_LogicChType v_TCHType;
    G_L2.send (cas_G_L2_DATA_REQ (p_CellId, 0, tsc_PhyCh0, p_ChannelType, p_SubChannel, cs_G_RFN_Omit, cs_G_SetupMTCall ( cs_BcapSpeechMT )));
    G_L2.receive (car_G_L2_DATA_IND ( p_CellId, ?, ? ,?, ?, ?,  cr_G_CallConfirmed (?,?)));
    v_TCHType := f_GERAN_EnterU10_MT_Step12_19 (p_CellId, p_ChannelType, p_SubChannel);
    return v_TCHType;
  }


After:
	function f_GERAN_EnterU10_MT_Step10_19 (GERAN_CellId_Type p_CellId,
                                          G_LogicChType p_ChannelType,
                                          SubChannelNumber p_SubChannel) runs on GERAN_PTC return G_LogicChType
  {
    var G_LogicChType v_TCHType;
    G_L2.send (cas_G_L2_DATA_REQ (p_CellId, 0, tsc_PhyCh0, p_ChannelType, p_SubChannel, cs_G_RFN_Omit, cs_G_SetupMTCall ( cs_BcapSpeechMT )));
    //Stream Identifier is optional. 
    G_L2.receive (car_G_L2_DATA_IND ( p_CellId, ?, ? ,?, ?, ?,  cr_G_CallConfirmed (?)));
    v_TCHType := f_GERAN_EnterU10_MT_Step12_19 (p_CellId, p_ChannelType, p_SubChannel);
    return v_TCHType;
  }


Change 3 – Correction to function ‘f_GERAN_EnterU10_MT_Step12_19()’ 
	Function name
	f_GERAN_EnterU10_MT_Step12_19

	Reason for change
	The Transaction Identifier flag in function f_GERAN_EnterU10_MT_Step12_19 should be ‘0’ B since the SS is the initiator of the transaction for the MT call.

	Summary of change
	Used cs_TI_MT as the Transaction Identifier whilst sending CONNECT ACK.

	TTCN module
	GERAN_CommonFunctions.ttcn

	MCC160 Comment
	


Before:
	function f_GERAN_EnterU10_MT_Step12_19 (GERAN_CellId_Type p_CellId,
                                          G_LogicChType p_ChannelType,
                                          SubChannelNumber p_SubChannel) runs on GERAN_PTC return G_LogicChType
  {
    var G_LogicChType v_TCHType;
    var SubChannelNumber v_TCHSubChannel;
    var G_L2_DATAMESSAGE_IND v_EstInd, v_Connect;
    var TI v_Ti_R, v_Ti_S;// Transaction Identifier. r = value used to receive CC PDU. s = value used to send CC PDU.
    var template (value) L2DataReqType v_AssignmentCmd := cs_G_AssignmentCommand(cs_G_ChModeSpeechFROrHRV1,
                                                                                 cs_G_ChannelDescr ( '00001'B, tsc_TN1, f_GERAN_TCHFreq_Get(p_CellId), px_TSC ),
                                                                                 omit,
                                                                                 omit,
                                                                                 omit,
                                                                                 omit,
                                                                                 cs_CipherModeSettingGSM) ;
    G_L2.send (cas_G_L2_DATA_REQ (p_CellId, 0, tsc_PhyCh0, p_ChannelType, p_SubChannel, cs_G_RFN_Omit,  ( v_AssignmentCmd )));
    G_L2.receive (car_G_L2_Estab_IND (p_CellId, ?, ?, ?, omit)) // receive establish indication first
      -> value v_EstInd;
    v_TCHType := v_EstInd.sabm.g_LogicChType;
    v_TCHSubChannel := v_EstInd.sabm.subChannel;
    alt {
      []  G_L2.receive (car_G_L2_DATA_IND ( p_CellId, ?, tsc_TchCh, ?, ?, ?,  cr_G_AssignmentComplete)) {}
      []  G_L2.receive (car_G_L2_DATA_IND ( p_CellId, ?, ? , ?, ?, ?,  cr_G_AssignmentFailure)) {
        f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Assignment Failed");
      }
    }
    alt {
      []  G_L2.receive (car_G_L2_DATA_IND ( p_CellId, ?, ? ,?, ?, ?,  cr_G_Connect (cr_TI_Any)))
        -> value v_Connect{}
      []  G_L2.receive (car_G_L2_DATA_IND ( p_CellId, ?, ? ,?, ?, ?,  cr_G_Alerting (?))) {
        f_UT_AnswerCSCall (UT);
        G_L2.receive (car_G_L2_DATA_IND ( p_CellId, ?, ? ,?, ?, ?,  cr_G_Connect (cr_TI_Any)))
          -> value v_Connect;
      }
    }
    // release old channel
    G_L2.send (cas_G_L2_Release_REQ(p_CellId, tsc_PhyCh0, p_ChannelType, p_SubChannel));
    G_L2.receive (car_G_L2_Release_CNF);
    v_Ti_R := v_Connect.l3.msg.connectul.ti;

    v_Ti_S := v_Ti_R;

    v_Ti_S.tiVal := v_Ti_R.tiVal xor4b '1'B;

    G_L2.send (cas_G_L2_DATA_REQ (p_CellId, 0, tsc_TchCh, v_TCHType, v_TCHSubChannel, cs_G_RFN_Omit, cs_G_ConnectAck (v_Ti_S)));
    return v_TCHType;


After:
	function f_GERAN_EnterU10_MT_Step12_19 (GERAN_CellId_Type p_CellId,
                                          G_LogicChType p_ChannelType,
                                          SubChannelNumber p_SubChannel) runs on GERAN_PTC return G_LogicChType
  {
    var G_LogicChType v_TCHType;
    var SubChannelNumber v_TCHSubChannel;
    var G_L2_DATAMESSAGE_IND v_EstInd, v_Connect;
    var TI v_Ti_R, v_Ti_S;// Transaction Identifier. r = value used to receive CC PDU. s = value used to send CC PDU.
    var template (value) L2DataReqType v_AssignmentCmd := cs_G_AssignmentCommand(cs_G_ChModeSpeechFROrHRV1,
                                                                                 cs_G_ChannelDescr ( '00001'B, tsc_TN1, f_GERAN_TCHFreq_Get(p_CellId), px_TSC ),
                                                                                 omit,
                                                                                 omit,
                                                                                 omit,
                                                                                 omit,
                                                                                 cs_CipherModeSettingGSM) ;
    G_L2.send (cas_G_L2_DATA_REQ (p_CellId, 0, tsc_PhyCh0, p_ChannelType, p_SubChannel, cs_G_RFN_Omit,  ( v_AssignmentCmd )));
    G_L2.receive (car_G_L2_Estab_IND (p_CellId, ?, ?, ?, omit)) // receive establish indication first
      -> value v_EstInd;
    v_TCHType := v_EstInd.sabm.g_LogicChType;
    v_TCHSubChannel := v_EstInd.sabm.subChannel;
    alt {
      []  G_L2.receive (car_G_L2_DATA_IND ( p_CellId, ?, tsc_TchCh, ?, ?, ?,  cr_G_AssignmentComplete)) {}
      []  G_L2.receive (car_G_L2_DATA_IND ( p_CellId, ?, ? , ?, ?, ?,  cr_G_AssignmentFailure)) {
        f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Assignment Failed");
      }
    }
    alt {
      []  G_L2.receive (car_G_L2_DATA_IND ( p_CellId, ?, ? ,?, ?, ?,  cr_G_Connect (cr_TI_Any)))
        -> value v_Connect{}
      []  G_L2.receive (car_G_L2_DATA_IND ( p_CellId, ?, ? ,?, ?, ?,  cr_G_Alerting (?))) {
        f_UT_AnswerCSCall (UT);
        G_L2.receive (car_G_L2_DATA_IND ( p_CellId, ?, ? ,?, ?, ?,  cr_G_Connect (cr_TI_Any)))
          -> value v_Connect;
      }
    }
    // release old channel
    G_L2.send (cas_G_L2_Release_REQ(p_CellId, tsc_PhyCh0, p_ChannelType, p_SubChannel));
    G_L2.receive (car_G_L2_Release_CNF);
    G_L2.send (cas_G_L2_DATA_REQ (p_CellId, 0, tsc_TchCh, v_TCHType, v_TCHSubChannel, cs_G_RFN_Omit, cs_G_ConnectAck ( cs_TI_MT )));
    return v_TCHType;


Change 4 – Correction to function ‘f_G_ChannelRelease()’ 
	Function name
	f_G_ChannelRelease

	Reason for change
	The TCH subchannel has been hardcoded to 15. Hence, in function f_G_ChannelRelease, the default value for parameter p_SubChannel should be set to 15.

	Summary of change
	Modified parameter p_SubChannel in function f_G_ChannelRelease to have default value of 15.

	TTCN module
	GERAN_CommonFunctions.ttcn

	MCC160 Comment
	


Before:
	function f_G_ChannelRelease (GERAN_CellId_Type p_CellId,
                               PhysicalChId p_PhyCh,
                               G_LogicChType p_ChannelType,
                               SubChannelNumber p_SubChannel) runs on GERAN_PTC
  {
    G_L2.send (cas_G_L2_DATA_REQ ( p_CellId, 0, p_PhyCh, p_ChannelType, p_SubChannel, cs_G_RFN_Omit, cs_G_ChRelease));
  } // end of f_G_ChannelRelease


After:
	function f_G_ChannelRelease (GERAN_CellId_Type p_CellId,
                               PhysicalChId p_PhyCh,
                               G_LogicChType p_ChannelType,
                               SubChannelNumber p_SubChannel := 15) runs on GERAN_PTC
  {
    G_L2.send (cas_G_L2_DATA_REQ ( p_CellId, 0, p_PhyCh, p_ChannelType, p_SubChannel, cs_G_RFN_Omit, cs_G_ChRelease));
  } // end of f_G_ChannelRelease


Change 5 – Correction to function ‘f_GERAN_SS_CellRelease()’ 
	Function name
	f_GERAN_SS_CellRelease

	Reason for change
	In function f_GERAN_SS_CellRelease, the default value of parameter p_TCH is set to FALSE. The condition to decide whether TCH has to be deleted should be based upon parameter p_TCH being TRUE rather than FALSE. Otherwise, there would be no meaning to the variable v_IsTCH being set to TRUE in function f_GERAN_Postamble.

	Summary of change
	Changed the condition to check if parameter p_TCH is TRUE before deleting the  TCH.

	TTCN module
	GERAN_CommonFunctions.ttcn

	MCC160 Comment
	


Before:
	function f_GERAN_SS_CellRelease (GERAN_CellId_Type p_CellId,
                                   boolean p_MinUL := false,
                                   boolean p_TCH := false) runs on GERAN_PTC
  { /* @sic R5s120050 MCC160 implementation sic@ */
    f_GERAN_CPHY_CONFIG_Common(cas_G_CL1_DeleteChannel_REQ ( p_CellId, tsc_PhyCh0));
    if (not p_MinUL) {
      f_GERAN_CPHY_CONFIG_Common(cas_G_CL1_DeleteChannel_REQ ( p_CellId, tsc_PhyCh1));
    }
    if (not p_TCH) { // @sic R5s120498 sic@
       f_GERAN_CPHY_CONFIG_Common(cas_G_CL1_DeleteChannel_REQ ( p_CellId, tsc_TchCh));
    }
    f_GERAN_CPHY_CONFIG_Common(cas_G_CL1_DeleteCell_REQ ( p_CellId ));
  } //end of f_GPRS_SS_CellRelease


After:
	function f_GERAN_SS_CellRelease (GERAN_CellId_Type p_CellId,
                                   boolean p_MinUL := false,
                                   boolean p_TCH := false) runs on GERAN_PTC
  { /* @sic R5s120050 MCC160 implementation sic@ */
    f_GERAN_CPHY_CONFIG_Common(cas_G_CL1_DeleteChannel_REQ ( p_CellId, tsc_PhyCh0));
    if (not p_MinUL) {
      f_GERAN_CPHY_CONFIG_Common(cas_G_CL1_DeleteChannel_REQ ( p_CellId, tsc_PhyCh1));
    }
    if ( p_TCH ) { // @sic R5s120498 sic@
       f_GERAN_CPHY_CONFIG_Common(cas_G_CL1_DeleteChannel_REQ ( p_CellId, tsc_TchCh));
    }
    f_GERAN_CPHY_CONFIG_Common(cas_G_CL1_DeleteCell_REQ ( p_CellId ));
  } //end of f_GPRS_SS_CellRelease


Change 6 – Correction to function ‘f_GERAN_CSFB_Redirection_CCO_MO_Steps3_32()’ 
	Function name
	f_GERAN_CSFB_Redirection_CCO_MO_Steps3_32

	Reason for change
	In function f_GERAN_CSFB_Redirection_CCO_MO_Steps3_32, the Logical Channel type value returned by function f_GERAN_EnterU10_MO_Step5_19 should be used while calling functions f_GERAN_CC_Disconnect and f_G_ChannelRelease.

	Summary of change
	Declared a variable v_TCHType to capture the return value and passed it as a parameter in functions f_GERAN_CC_Disconnect and f_G_ChannelRelease.

	TTCN module
	MultiLayer_Procedures_GERAN.ttcn

	MCC160 Comment
	


Before:
	function f_GERAN_CSFB_Redirection_CCO_M0_Steps3_32 (GERAN_CellId_Type p_CellId) runs on GERAN_PTC
  {
    var G_L2_DATAMESSAGE_IND v_G_L2_DATAMESSAGE_IND;
    var G_LogicChType v_ChannelType;
    var SubChannelNumber v_SubChannel;
    //Receive GPRS suspension request
    G_L2.receive (car_G_L2_DATA_IND (p_CellId,
                                     ?,
                                     tsc_PhyCh0,
                                     ?,
                                     ?,
                                     ?,
                                     cr_GPRS_SuspensionReq(?, cr_RAI (omit),? ))); 

    //Receive CM Service Request
    G_L2.receive (car_G_L2_Estab_IND (p_CellId,
                                      ?,
                                      ?,
                                      ?,
                                      cr_G_CMServiceRequest (?))) -> value v_G_L2_DATAMESSAGE_IND;
    v_ChannelType := v_G_L2_DATAMESSAGE_IND.sabm.g_LogicChType;
    v_SubChannel := v_G_L2_DATAMESSAGE_IND.sabm.subChannel;
    f_GERAN_PreliminaryPass (__FILE__, __LINE__, "CS fallback to GERAN with redirection or CCO / MO call (DTM not supported) Step 4");
    // @sic R5s120339 sic@
    G_L2.receive (car_G_L2_DATA_IND (p_CellId, ?, tsc_PhyCh0, ?, ?, ?, cr_G_ClassmarkChangeAny));
    G_L2.receive (car_G_L2_DATA_IND (p_CellId, ?, tsc_PhyCh0, ?, ?, ?, cr_G_UtranClassmarkChange));
    f_GERAN_EnterU10_MO_Steps5_19(p_CellId, v_ChannelType, v_SubChannel);
    //Steps 20-22: Disconnect the MT call with MT procedure
    f_GERAN_CC_Disconnect(p_CellId, v_ChannelType, v_ChannelType);
    //Release channel
    f_G_ChannelRelease (p_CellId, tsc_PhyCh0, v_ChannelType, v_ChannelType);
    //Perform RAU
    f_GERAN_508CheckCampOnNewGeranCell(p_CellId);
  }


After:
	function f_GERAN_CSFB_Redirection_CCO_M0_Steps3_32 (GERAN_CellId_Type p_CellId) runs on GERAN_PTC
  {
    var G_L2_DATAMESSAGE_IND v_G_L2_DATAMESSAGE_IND;
    var G_LogicChType v_ChannelType, v_TCHType;
    var SubChannelNumber v_SubChannel;
    //Receive CM Service Request
    G_L2.receive (car_G_L2_Estab_IND (p_CellId,
                                      ?,
                                      ?,
                                      ?,
                                      cr_G_CMServiceRequest (?))) -> value v_G_L2_DATAMESSAGE_IND;
    v_ChannelType := v_G_L2_DATAMESSAGE_IND.sabm.g_LogicChType;
    v_SubChannel := v_G_L2_DATAMESSAGE_IND.sabm.subChannel;
    f_GERAN_PreliminaryPass (__FILE__, __LINE__, "CS fallback to GERAN with redirection or CCO / MO call (DTM not supported) Step 4");
    // @sic R5s120339 sic@
    G_L2.receive (car_G_L2_DATA_IND (p_CellId, ?, tsc_PhyCh0, ?, ?, ?, cr_G_ClassmarkChangeAny));
    G_L2.receive (car_G_L2_DATA_IND (p_CellId, ?, tsc_PhyCh0, ?, ?, ?, cr_G_UtranClassmarkChange));
    //Receive GPRS suspension request
    G_L2.receive (car_G_L2_DATA_IND (p_CellId,
                                     ?,
                                     tsc_PhyCh0,
                                     ?,
                                     ?,
                                     ?,
                                     cr_GPRS_SuspensionReq(?, cr_RAI (omit),? ))); // @sic R5s120509 sic@
    v_TCHType := f_GERAN_EnterU10_MO_Steps5_19(p_CellId, v_ChannelType, v_SubChannel);
    //Steps 20-22: Disconnect the MO call with MO procedure
    f_GERAN_CC_Disconnect(p_CellId, v_TCHType, -, cs_TI_MO);
    //Release channel
    f_G_ChannelRelease (p_CellId, tsc_TchCh, v_TCHType);
    //Perform RAU
    f_GERAN_508CheckCampOnNewGeranCell(p_CellId);
  }


Change 7 – Correction to function ‘f_GERAN_EnterU10_MO_Step5_19()’ 
	Function name
	f_GERAN_EnterU10_MO_Step5_19

	Reason for change
	In function f_GERAN_EnterU10_MO_Step5_19, the CONNECT signalling message has to be sent on the TCH Physical ChId.

	Summary of change
	Used tsc_TchCh as the PhysicalChId whilst sending the CONNECT signalling message.

	TTCN module
	GERAN_CommonFunctions.ttcn

	MCC160 Comment
	


Before:
	function f_GERAN_EnterU10_MO_Steps5_19(GERAN_CellId_Type p_CellId,
                                         G_LogicChType     p_ChannelType,
                                         SubChannelNumber  p_SubChannel) runs on GERAN_PTC return G_LogicChType
  {
    var G_L2_DATAMESSAGE_IND v_SetUp, v_EstInd;
    var TI v_Ti_S; //Sending Transaction Identifier
    var G_LogicChType v_TCHType;
    var SubChannelNumber v_TCHSubChannel;
    var template (value) L2DataReqType v_AssignmentCmd := cs_G_AssignmentCommand(cs_G_ChModeSpeechFROrHRV1,
                                                                                 cs_G_ChannelDescr ( '00001'B, tsc_TN1, f_GERAN_TCHFreq_Get(p_CellId), px_TSC ),
                                                                                 omit,
                                                                                 omit,
                                                                                 omit,
                                                                                 omit,
                                                                                 cs_CipherModeSettingGSM);
    //@siclog "Step 5-6" siclog@
    //AUTHENTICATION REQUEST and AUTHENTICATION RESPONSE
    f_GERAN_MM_Authentication (p_CellId, p_ChannelType, p_SubChannel);
    //@siclog "Step 7-9" siclog@
    //CIPHERING MODE COMMAND & CIPHERING MODE COMPLETE
    f_GERAN_Ciphering_Mode_Setting (p_CellId, tsc_PhyCh0, p_ChannelType, p_SubChannel);
    //@siclog "Step 10" siclog@
    //Receive Set Up message
    G_L2.receive (car_G_L2_DATA_IND ( p_CellId, ?, ? ,?, ?, ?,  cr_G_SetupUL_MO)) -> value v_SetUp;
    v_Ti_S := v_SetUp.l3.msg.setupul.ti;
    v_Ti_S.tiFlag := '1'B;
    //@siclog "Step 11" siclog@
    //Send call proceeding
    G_L2.send (cas_G_L2_DATA_REQ (p_CellId, 0, tsc_PhyCh0, p_ChannelType, p_SubChannel, cs_G_RFN_Omit, cs_G_CallProc( v_Ti_S )));
    //@siclog "Step 12" siclog@
    //Send Alert message
    G_L2.send (cas_G_L2_DATA_REQ (p_CellId, 0, tsc_PhyCh0, p_ChannelType, p_SubChannel, cs_G_RFN_Omit, cs_G_Alert(v_Ti_S)));
    //@siclog "Step 13" siclog@
    f_UT_AnswerCSCall (UT);
    //@siclog "Step 14-15" siclog@
    //Assignment of Traffic Channel
    G_L2.send (cas_G_L2_DATA_REQ (p_CellId, 0, tsc_PhyCh0, p_ChannelType, p_SubChannel, cs_G_RFN_Omit,  (v_AssignmentCmd)));
    G_L2.receive (car_G_L2_Estab_IND (p_CellId, ?, ?, ?, omit)) // receive establish indication first
      -> value v_EstInd;
    v_TCHType := v_EstInd.sabm.g_LogicChType;
    v_TCHSubChannel := v_EstInd.sabm.subChannel;
    alt {
      [] G_L2.receive (car_G_L2_DATA_IND ( p_CellId, ?, tsc_TchCh,?, ?, ?,  cr_G_AssignmentComplete)) {}
      [] G_L2.receive (car_G_L2_DATA_IND ( p_CellId, ?, ? ,?, ?, ?,  cr_G_AssignmentFailure)) {
        f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Assignment Failed");
      }
    }
    //@siclog "Step 16" siclog@
    //Send Connect message
    G_L2.send (cas_G_L2_DATA_REQ (p_CellId, 0, tsc_PhyCh0, v_TCHType, v_TCHSubChannel, cs_G_RFN_Omit, cs_G_Connect(v_Ti_S)));
    //@siclog "Step 17" siclog@
    //Receive CONNECT ACKNOWLEDGE
    G_L2.receive (car_G_L2_DATA_IND (p_CellId, ?, ?, ?, ?, ?, cr_G_ConnAck(cr_TI_Any)));
    return v_TCHType;
  }


After:
	function f_GERAN_EnterU10_MO_Steps5_19(GERAN_CellId_Type p_CellId,
                                         G_LogicChType     p_ChannelType,
                                         SubChannelNumber  p_SubChannel) runs on GERAN_PTC return G_LogicChType
  {
    var G_L2_DATAMESSAGE_IND v_SetUp, v_EstInd;
    var TI v_Ti_S; //Sending Transaction Identifier
    var G_LogicChType v_TCHType;
    var SubChannelNumber v_TCHSubChannel;
    var template (value) L2DataReqType v_AssignmentCmd := cs_G_AssignmentCommand(cs_G_ChModeSpeechFROrHRV1,
                                                                                 cs_G_ChannelDescr ( '00001'B, tsc_TN1, f_GERAN_TCHFreq_Get(p_CellId), px_TSC ),
                                                                                 omit,
                                                                                 omit,
                                                                                 omit,
                                                                                 omit,
                                                                                 cs_CipherModeSettingGSM);
    //@siclog "Step 5-6" siclog@
    //AUTHENTICATION REQUEST and AUTHENTICATION RESPONSE
    f_GERAN_MM_Authentication (p_CellId, p_ChannelType, p_SubChannel);
    //@siclog "Step 7-9" siclog@
    //CIPHERING MODE COMMAND & CIPHERING MODE COMPLETE
    f_GERAN_Ciphering_Mode_Setting (p_CellId, tsc_PhyCh0, p_ChannelType, p_SubChannel);
    //@siclog "Step 10" siclog@
    //Receive Set Up message
    G_L2.receive (car_G_L2_DATA_IND ( p_CellId, ?, ? ,?, ?, ?,  cr_G_SetupUL_MO)) -> value v_SetUp;
    v_Ti_S := v_SetUp.l3.msg.setupul.ti;
    v_Ti_S.tiFlag := '1'B;
    //@siclog "Step 11" siclog@
    //Send call proceeding
    G_L2.send (cas_G_L2_DATA_REQ (p_CellId, 0, tsc_PhyCh0, p_ChannelType, p_SubChannel, cs_G_RFN_Omit, cs_G_CallProc( v_Ti_S )));
    //@siclog "Step 12" siclog@
    //Send Alert message
    G_L2.send (cas_G_L2_DATA_REQ (p_CellId, 0, tsc_PhyCh0, p_ChannelType, p_SubChannel, cs_G_RFN_Omit, cs_G_Alert(v_Ti_S)));
    //@siclog "Step 13" siclog@
    f_UT_AnswerCSCall (UT);
    //@siclog "Step 14-15" siclog@
    //Assignment of Traffic Channel
    G_L2.send (cas_G_L2_DATA_REQ (p_CellId, 0, tsc_PhyCh0, p_ChannelType, p_SubChannel, cs_G_RFN_Omit,  (v_AssignmentCmd)));
    G_L2.receive (car_G_L2_Estab_IND (p_CellId, ?, ?, ?, omit)) // receive establish indication first
      -> value v_EstInd;
    v_TCHType := v_EstInd.sabm.g_LogicChType;
    v_TCHSubChannel := v_EstInd.sabm.subChannel;
    alt {
      [] G_L2.receive (car_G_L2_DATA_IND ( p_CellId, ?, tsc_TchCh,?, ?, ?,  cr_G_AssignmentComplete)) {}
      [] G_L2.receive (car_G_L2_DATA_IND ( p_CellId, ?, ? ,?, ?, ?,  cr_G_AssignmentFailure)) {
        f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Assignment Failed");
      }
    }
    //@siclog "Step 16" siclog@
    //Send Connect message
    G_L2.send (cas_G_L2_DATA_REQ (p_CellId, 0, tsc_TchCh, v_TCHType, v_TCHSubChannel, cs_G_RFN_Omit, cs_G_Connect(v_Ti_S)));
    //@siclog "Step 17" siclog@
    //Receive CONNECT ACKNOWLEDGE
    G_L2.receive (car_G_L2_DATA_IND (p_CellId, ?, ?, ?, ?, ?, cr_G_ConnAck(cr_TI_Any)));
    return v_TCHType;
  }


Change 8 – Correction to template ‘cs_ChMode’ 
	Template name
	cs_ChMode

	Reason for change
	Record ChMode has the element ‘iei’ as a IEI8_Type which is of type OCTETSTRING. However, in TTCN-2, ChMode has the ‘iei’ as being defined as type BITSTRING. Hence, there is a mis-alignment between TTCN-3 and TTCN-2.

	Summary of change
	Modified record ChMode to change the ‘iei’ type to B8_Type instead of IEI8_Type. Corresponding changes done to templates where record ChMode has been used.

	TTCN module
	GERAN_Templates.ttcn, GERAN_TypeDefs.ttcn

	MCC160 Comment
	


Before:
	type record ChMode {
  IEI8_Type              iei    optional,   //'01100011'B
  B8_Type              mode
};

template (value) ChMode cs_ChMode(template (omit) IEI8_Type p_IEI := '63'O, B8_Type p_Mode) :=
  { /* @status    APPROVED */
    iei             := p_IEI, // @sic R5s120498 sic@
    mode            := p_Mode
  };

template (value) ChMode cs_ChMode cs_G_ChModeSpeechFROrHRV1 := cs_ChMode('63'O, '00000001'B);




After:
	type record ChMode {
  B8_Type   iei    optional, //changing the type definitions '01100011'B
  B8_Type   mode
};

template (value) ChMode cs_ChMode(template (omit) B8_Type p_IEI := '01100011'B, B8_Type p_Mode) :=
  { /* @status    APPROVED */
    iei             := p_IEI, // @sic R5s120498 sic@
    mode            := p_Mode
  };

template (value) ChMode cs_ChMode cs_G_ChModeSpeechFROrHRV1 := cs_ChMode('01100011'B, '00000001'B);

